# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Daniel Wallace <danielwallace at gtmanfred dot com>

pkgname=python-cinderclient
pkgver=9.4.0
_commit=f1f14dfbb77dd24e2a9c8688bb6ec688d0f8229c
pkgrel=1
pkgdesc="OpenStack Block Storage API Client Library"
arch=('any')
url="https://docs.openstack.org/developer/python-cinderclient"
license=('Apache')
depends=('python-pbr' 'python-prettytable' 'python-keystoneauth1' 'python-oslo-i18n'
         'python-oslo-utils' 'python-requests' 'python-stevedore')
makedepends=('git')
checkdepends=('python-ddt' 'python-fixtures' 'python-requests-mock' 'python-testtools'
              'python-stestr' 'python-oslo-serialization')
source=("git+https://github.com/openstack/python-cinderclient.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd python-cinderclient
  python setup.py build
}

check() {
  cd python-cinderclient
  stestr run
}

package() {
  cd python-cinderclient
  python setup.py install --root="$pkgdir" --optimize=1
}
